

A. Les opérateurs arithmétiques et les formules

1. Les règles de priorité des opérateurs

Les opérateurs arithmétiques d'Excel sont au nombre de 6 :

- ▶ + (addition)
- ▶ - (soustraction ou négation)
- ▶ / (division)
- ▶ ^ (puissance)
- ▶ % (pourcentage)

Il est possible également d'utiliser les parenthèses (). L'ordre de priorité des opérateurs est le suivant (du plus prioritaire au moins prioritaire).

- ▶ - (négation)
- ▶ % (pourcentage)
- ▶ ^ (puissance)
- ▶ * et / (multiplication et division)
- ▶ + et - (addition et soustraction)

L'évaluation des priorités se fait toujours à l'intérieur des parenthèses.

Les formules commencent toujours par le signe =.

2. Exemples de formules

Les exemples présentés sont tous tirés de formules de **mathématiques financières**. Celles-ci présentent l'avantage de pouvoir être écrites uniquement à l'aide d'opérateurs arithmétiques.

Exemple 1

Le taux mensuel équivalent

t_a est le taux annuel et t_{me} le taux mensuel équivalent.

$$t_{me} = (1 + t_a)^{\frac{1}{12}} - 1$$

B2		fx		=(1+B1)^(1/12)-1
	A	B	E	
1	t_a :	2,65%	Formule	
2	t_{me} :	0,218%	=(1+B1)^(1/12)-1	

Exemple 2

Valeur V_n acquise par un capital K placé à un taux annuel t_a pendant n années.

$$V_n = K \cdot (1 + t_a)^n$$

Application

10 000 € placés pendant 10 ans au taux annuel de 2,25 %.

B7		fx		=B4*(1+B6)^B5
	A	B	E	
4	K :	10 000		
5	n :	5		
6	t_a :	2,25%	Formule	
7	V_n :	11 176,78	=B4*(1+B6)^B5	

Exemple 3

Valeur future V_n d'une suite de versements constants a effectués pendant n années et placés à un taux annuel t_a .

$$V_n = a \cdot \frac{(1+t_a)^n - 1}{t_a}$$

Application

5.000 € versés annuellement pendant 8 ans et placés au taux annuel de 1,75 %.

		B12	
		=B9*((1+B11)^B10-1)/B11	
	A	B	E
9	a :	5 000,00	
10	n :	8	
11	t _a :	1,75%	Formule
12	V _n :	42 537,65	=B9*((1+B11)^B10-1)/B11

Exemple 4

Calcul de la mensualité m de remboursement d'un emprunt au taux annuel t_a d'un montant K pendant n années (la formule inclut le calcul du taux mensuel équivalent).

$$m = K \cdot \frac{(1+t_a)^{\frac{1}{12}} - 1}{1 - (1+t_a)^{-n}}$$

Application

Mensualité de remboursement d'un emprunt de 10.000 € sur 5 ans au taux annuel de 2,50 %.

		B18	
		=B15*((1+B17)^(1/12)-1)/(1-(1+B17)^-B16)	
	A	B	E
15	K :	10 000	
16	n :	5	
17	t _a :	2,50%	Formule
18	m :	177,35	=B15*((1+B17)^(1/12)-1)/(1-(1+B17)^-B16)

Exemple 5

Calcul du capital V_k restant à rembourser après paiement de la $k^{\text{ème}}$ mensualité d'un emprunt de montant K sur n années au taux annuel t_a .

$$V_k = K \cdot \frac{1 - (1+t_a)^{-\frac{12n-k}{12}}}{1 - (1+t_a)^{-n}}$$

Application

Capital restant à rembourser après remboursement de la 3^e mensualité d'un emprunt de 100 000 € sur 10 ans au taux annuel de 2,75 %.

		=B21*(1-(1+B24)^(-(12*B22-B23)/12))/(1-(1+B24)^-B22)	
	A	B	E
21	K :	100 000	
22	n :	10	
23	k :	3	
24	t _a :	2,75%	Formule
25	V _k :	97 816,40	=B21*(1-(1+B24)^(-(12*B22-B23)/12))/(1-(1+B24)^-B22)

B. Les fonctions de base

Dans cette partie sont présentées trois fonctions importantes d'Excel. Leur usage répond à des besoins fréquents. Il s'agit des fonctions SOMME, SI et RECHERCHEV.

1. La fonction SOMME

La fonction SOMME a pour argument(s) une (ou des) plage(s) de cellules. Elle présente l'avantage, par rapport à l'addition simple, d'éliminer les cellules vides et les cellules contenant du texte.

	A	B	E
30	Agence	Volume	
31	Paris	150	
32	Nantes		
33	Bordeaux	Travaux	
34	Lille	100	Formule
35	TOTAL	250	=SOMME(B31:B34)

La formule avec la fonction SOMME donne un résultat correct, alors que la formule d'addition : =B31+B32+B33+B34 génère l'erreur #VALEUR!

2. La fonction SI

a. Syntaxe de base

La fonction SI permet, sur condition, d'affecter à une cellule un contenu choisi entre deux possibilités. Ce contenu peut être un nombre, un texte ou une formule. Il faut noter que ce résultat peut elle-même être une autre fonction SI.

La syntaxe de la fonction est la suivante :

=SI(condition;contenu quand vrai;contenu quand faux)

Exemple 1

Dans une gestion commerciale, chaque client bénéficie en fin de période d'une ristourne calculée par l'application d'un taux au volume d'affaires réalisé pendant la période.

Si le volume atteint ou dépasse 10000, le client bénéficie du taux de ristourne majoré, mais du taux de ristourne de base dans le cas contraire.

Les cellules B37 et B38 sont nommées respectivement **Taux_Ristourne_Base** et **Taux_Ristourne_Majorée**.

C41		=SI(B41>=10000;Taux_Ristourne_Majorée;Taux_Ristourne_Base)	
	A	B	C
37	Rist. Base :	1,00%	
38	Rist. Majorée:	3,50%	
39			
40	Client	Volume	Rist.
41	Client 1	15 000	3,50%
42	Client 2	8 000	1,00%
43	Client 3	9 900	1,00%
44	Client 4	10 000	1,00%
45	Client 5	8 600	1,00%
46	Client 6	14 400	3,50%

La formule de la cellule C41 est recopiée telle quelle sur la plage C42:C46.

b. L'expression des conditions

Les conditions simples

Une condition simple comporte trois éléments : la valeur à comparer, l'opérateur de comparaison et la valeur de comparaison.

Valeur à comparer	Opérateur	Valeur de comparaison
Cellule	=	Nombre
Expression	<	Texte
	<=	Cellule
	>	Expression
	>=	
	<>	



Lorsque des textes sont utilisés dans des formules Excel, ils doivent obligatoirement être placés entre guillemets (""). Cette convention permet à Excel de distinguer les textes des noms de cellules.

Toutes les combinaisons entre ces 3 éléments sont possibles.

Voici quelques exemples de conditions simples valides.

=SI(B7>=C7;...;...)

Si le contenu de la cellule B7 est supérieur ou égal au contenu de la cellule C7.

=SI(H8<>"TOTAL";...;...)

Si la cellule H8 ne contient pas le mot "TOTAL".

=SI(SOMME(B2:B500)>100000;...;...)

Si la somme des contenus des cellules de la plage B2:B500 est supérieure à 100000.

=SI(SOMME(B2:B500)<>SOMME(C2:C500);...;...)

Si la somme des contenus des cellules de la plage B2:B500 est différente de la somme des contenus des cellules de la plage C2:C500.



Une condition simple a pour résultat une valeur booléenne (VRAI ou FAUX). Réciproquement, toute valeur booléenne obtenue par calcul peut servir de condition dans une fonction SI. Excel comporte des fonctions à résultat booléen : EST.PAIR(), EST.IMPAIR(), ESTTEXTE(), ESTNUM(), ESTERREUR(), etc. Par exemple, la syntaxe : =SI(EST.IMPAIR(B5);...;...) est correcte.

Les conditions composées

Les conditions composées sont des combinaisons de plusieurs autres conditions par les **opérateurs logiques** ET, OU et NON. Ces autres conditions peuvent être elles-mêmes des conditions simples ou d'autres conditions composées.

L'opérateur logique ET

Pour que le résultat de la condition composée par ET soit égal à VRAI, toutes les conditions qui la composent doivent elles-mêmes être égales à VRAI. L'opérateur s'exprime par la fonction Excel ET(cond1;cond2;cond3;...).

L'opérateur logique OU

Pour que le résultat de la condition composée par OU soit égal à VRAI, au moins une des conditions qui la composent doit être égale à VRAI. L'opérateur s'exprime par la fonction Excel OU(cond1;cond2;cond3;...).

L'opérateur logique NON

Cet opérateur a pour effet d'inverser la condition. VRAI est transformé en FAUX et vice-versa. Il s'exprime par la fonction NON (condition).